Evaluating the Alignment of Astronomical Linear Polarization Data, introductory software

Richard Shurtleff *

Abstract

This article is a Mathematica notebook that is meant to serve as a template. User-supplied astronomical observations of transverse vectors on the sky can be evaluated, their alignment judged by the so-called Hub test. The test can be applied to any set of transverse vectors on a spherical surface, but the language here applies to linear polarization directions of electromagnetic radiation from astronomical sources. This article presents a simulation, analyzing artificial data as an illustration of the process. The analysis produces a numerical value quantifying the alignment and its significance. A visual representation of the alignment is developed, mapping regions of convergence and divergence on the Celestial sphere.

Keywords: Polarization ; Alignment ; Computer Program

*Department of Sciences, Wentworth Institute of Technology, 550 Huntington Avenue, Boston, MA, USA, 02115, orcid.org/0000-0001-5920-759X, e-mail addresses: shurtleffr@wit.edu, momentummatrix@yahoo.com}}

0. Preface

This notebook is intended to be used as a template. In order to use the notebook, it must be recognized and interpreted by Wolfram Mathematica's proprietary software. Since this file is published as a pdf, it must be somehow translated into the Mathematica computer language. You can simply copy the text here keystroke-by-keystroke into an active Mathematica notebook. A link[0] to the Mathematica notebook is provided in the references.

Replace the simulated data in *Sec. 3* and run the notebook. One needs the location of the sources on the sky and a position angle at each source.

Transverse vectors on the sky can be observed for many situations, linear polarization, major/minor axes, jets and others. These observed asymmetries may be analyzed for their mutual alignment.

The work is based on an article[1] "Indirect polarization alignment with points on the sky, the Hub Test". This notebook was created using Wolfram Mathematica[2], Version Number: 12.1 which is running on Microsoft Windows(64-bit).

CONTENTS

1. Introduction

Given a collection of astronomical sources with linearly polarized electromagnetic emissions, one may ask whether the polarization directions align.

The Hub test answers the question of alignment indirectly. Instead of attempting to find direct correlations of the polarization directions of a number of sources, an alternative process is applied.

The basic idea is illustrated in the figures below. The Celestial sphere is pictured on the left and on the right is the plane tangent to the sphere at the source S. The linear polarization direction $\hat{v}_\psi$ lies in the tangent plane and determines the purple great circle on the sphere. A point H on the sphere and the location S of the source determine a second great circle, the blue circle drawn on the sphere at the left. Clearly, H and S must be distinct points on the sphere. The angle η, with $0° \leq \eta \leq 90°$, measures the "alignment of the polarization direction with the point H." Perfect alignment occurs when $\eta = 0°$ and the two great circles form a single circle.

The basic concept includes "avoidance", as well as alignment. Avoidance is high when the two directions $\hat{v}_\psi$ and $\hat{v}_H$ differ by a large angle, $\eta \to 90°$. Perpendicular great circles at S, $\eta = 90°$, would indicate the maximum avoidance of the polarization direction and the point on the sphere.

Out[◦]=



With many sources S_i, $i = 1, ..., N$, there are N alignment angles η_{iH} for the point H. To quantify the alignment of the N sources with the point H, calculate the arithmetic average alignment angle at H,

$$\overline{\eta}(\mathrm{H}) = \frac{1}{N} \sum_{i=1}^{N} \eta_{iH} \,. \qquad (1)$$

The alignment angle $\overline{\eta}$(H) is a function of position H on the sphere. The polarization directions are best aligned with the hub point H_{min} where the alignment angle is a minimum $\overline{\eta}_{min}$. The polarization directions most avoid the hub point H_{max} where the function $\overline{\eta}$(H) takes its maximum value $\overline{\eta}_{max}$. For a visual aid, see the map generated near the end of the notebook.

The Hub test is based on the idea that the polarization directions are well-aligned with each other when they are well-aligned with some point H_{min}. The point H_{max} is also distinguished by the collection of polarization directions; it is the most avoided point.

The hub test calculates $\overline{\eta}_{min}$ and $\overline{\eta}_{max}$ for a given collection of polarized sources. The smaller the value of $\overline{\eta}_{min}$, the better aligned the sources are. The larger the value of $\overline{\eta}_{max}$, the more significant their avoidance of H_{max}.

For more on the Hub test, see the article[1].

2. Preliminary

We work on a sphere in 3 dimensional Euclidean space. The sphere is called the “Celestial sphere” or simply the “sphere”. The center of the sphere is the origin of a 3D Cartesian coordinate system with coordinates (x, y,z) and the direction of the positive z -axis is “due North”. Right ascension, RA or α, and declination, dec or δ, are measured with respect to the direction of the positive x-axis, which has RA = 0° and dec = 0°.

From a point-of-view located outside the sphere, as in the figure in the Introduction, one pictures a source S plotted on the sphere so that local North is upward and local East is to the right. A “position angle” at the point S on the sphere is measured in the 2D plane tangent to the sphere at S. The position angle ψ is measured clockwise from local North with East to the right.

Definitions:

(α,δ)	Right Ascension RA and declination dec of a point on the sphere. Sometimes we use radians, sometimes degrees.
er(α,δ)	radial unit vector in a Cartesian coordinate system from the Origin to the point on the sphere with (RA,dec) = (α,δ), with α,δ in radians
eN(α,δ)	unit vector along local North at the point (α,δ) on the sphere, with α,δ in radians
eE(α,δ)	unit vector along local East at the point (α,δ) on the sphere, with α,δ in radians
αFROMr$(\hat{r})$	RA for the point on the sphere determined by radial unit vector $\hat{r}$, result in radians
δFROMr$(\hat{r})$	dec for the point on the sphere determined by radial unit vector $\hat{r}$, result in radians

```
In[1]:= er[α_, δ_] := er[α, δ] = {Cos[α] Cos[δ], Sin[α] Cos[δ], Sin[δ]} (* α,δ in radians *)
eN[α_, δ_] := eN[α, δ] = {-Cos[α] Sin[δ], -Sin[α] Sin[δ], Cos[δ]}
eE[α_, δ_] := eE[α, δ] = {-Sin[α], Cos[α], 0}
{"Check er.er = 1, er.eN = 0, er.eE = 0, eN.eN
   = 1, eN.eE = 0,eE.eE = 1, erXeE = eN, eEXeN = er, eNXer = eE: ",
 {0} == Union[Flatten[Simplify[{er[α, δ].er[α, δ] - 1, er[α, δ].eN[α, δ], er[α, δ].eE[α, δ],
      eN[α, δ].eN[α, δ] - 1, eN[α, δ].eE[α, δ], eE[α, δ].eE[α, δ] - 1, Cross[er[α, δ], eE[α, δ]] -
       eN[α, δ], Cross[eE[α, δ], eN[α, δ]] - er[α, δ], Cross[eN[α, δ], er[α, δ]] - eE[α, δ]}]]]}

Out[4]= {Check er.er = 1, er.eN = 0, er.eE = 0, eN.eN
   = 1, eN.eE = 0,eE.eE = 1, erXeE = eN, eEXeN = er, eNXer = eE: , True}
```

Get (α,δ) in radians from radial vector r, with $-\pi < \alpha < +\pi$ and $\frac{-\pi}{2} < \delta < \frac{+\pi}{2}$

```
In[5]:= αFROMr[r_] := N[ArcTan[Abs[r[[2]]/r[[1]]]]] /; (r[[2]] ≥ 0 && r[[1]] > 0)
αFROMr[r_] := N[π - ArcTan[Abs[r[[2]]/r[[1]]]]] /; (r[[2]] ≥ 0 && r[[1]] < 0)
αFROMr[r_] := N[-π + ArcTan[Abs[r[[2]]/r[[1]]]]] /; (r[[2]] < 0 && r[[1]] < 0)
αFROMr[r_] := N[-ArcTan[Abs[r[[2]]/r[[1]]]]] /; (r[[2]] < 0 && r[[1]] > 0)
αFROMr[r_] := π/2. /; (r[[2]] ≥ 0 && r[[1]] == 0)
αFROMr[r_] := -π/2. /; (r[[2]] < 0 && r[[1]] == 0)

In[11]:= δFROMr[r_] := N[ArcTan[r[[3]]/Sqrt[r[[1]]^2 + r[[2]]^2]]] /; (Sqrt[r[[1]]^2 + r[[2]]^2] > 0)
δFROMr[r_] := Sign[r[[3]]] π/2. /; (Sqrt[r[[1]]^2 + r[[2]]^2] == 0)
```

3. Input and Settings

This section is where you would enter your data for analysis. You can input source locations in various ways using the functions in Section 2 above.
Be careful of units. The angles α, δ, ψ are all expected to be in radians.

Definitions:

gridSpacing separation in degrees between grid points on a constant latitude circle and separation of constant latitude circles. There is no bunching at the poles.

ρRegion estimated radius of the region containing the sources, choose from ρRegion = {90° (whole sphere), 48°, 24°, 12°,5°, 0° (point-like)}.

nSrc number of sources in the region

αSrc Right Ascension (RA) at the sources, in radians

δSrc declinations (dec) at the sources, in radians

rSrc radial unit vectors in Cartesian coordinates from origin to sources S_i

ψn the polarization position angles for the EM radiation from the sources, in radians

dηContourPlot separation of successive contour lines on the map in Sec. 7, in degrees

mapDirectory folder on the computer where the map is to be saved

Settings

In[13]:=
```
gridSpacing = 2.(*, in degrees. This is a setting.*);
Print["The grid points are separated by ",
 gridSpacing, "° arcs along latitude and longitude."]
```

The grid points are separated by 2.° arcs along latitude and longitude.

In[15]:=
```
regionRadiusChoices = {90, 48, 24, 12, 5, 0};(*Do not change this statement*)
regionChoice = 3; (*This is a setting. The choice 24° is 3rd in the list. *)
rgnRadius = regionRadiusChoices[[regionChoice]];
Print["The region radius controls the constants ci and ai for statistics in Sec. 4."]
Print["The region radius ρ is set at ", rgnRadius, "°."]
```

The region radius controls the constants c_i and a_i for statistics in Sec. 4.

The region radius ρ is set at 24°.

In[20]:=
```
nSrc = 16;(*The number of sources. This is a setting.*)
```

In[21]:=
```
dηContourPlot = 4 ;(*, in degrees. This is a setting.*)
```

In[22]:=
```
mapDirectory =
  "C:\\Users\\shurt\\Dropbox\\HOME_DESKTOP-0MRE5OJ\\SendXXX_CJP_CEJPetc\\SendViXra\\
    20200715AlignmentMethod\\20200715AlignmentMMAnotebooks\\StarterKit";
(*This is a setting.*)
```

Inputs

In[23]:=
```
(*The locations of the sources Si.  Here (RA,dec) are the inputs and Cartesian
 coordinates are calculated. Alternatively, you can input rSrc and calculate αSrc,
δSrc with the functions αFROMr and δFROMr in Sec. 1.*)
αSrc = {1.0245, 0.2994, 0.8584, 0.4293, 0.7828, 0.7407, 1.1216, 0.5534,
   0.7863, 1.0897, 0.9064, 0.7216, 0.3302, 0.3788, 1.1390, 0.5709};(*Input*)
δSrc = {0.8400, 0.6266, 0.2472, 0.2780, 0.3821, 0.3826, 0.5953, 0.9090, 0.6663,
   0.6634, 0.4188, 0.6961, 0.5614, 0.7652, 0.8050, 0.2800}; (*Input*)
rSrc = Table[er[ αSrc[[i]], δSrc[[i]] ],  {i, nSrc}];(*calculated from Input.*)
```

In[26]:=
```
(*The polarization position angles for the
 EM radiation from the sources. This is an input.*)
ψn = {2.2816, 1.3406, 2.6725, 1.9480, 1.7352, 2.2421, 0.1986, 2.1445,
   2.3088, 2.0109, 1.6127, 0.3118, 1.6390, 2.3304, 2.4428, 1.8222};
```

4. Significance

When 5% or fewer results with random data are better then a result with observed data, the observed result is called "significant" by definition or by convention.
When 1% or fewer random results are better, then a result is called "very significant" by definition or by convention.

To determine the probability distributions and related formulas, we made many runs with random data and fit the results. There were 2000 runs for each combination of N sources in regions of radii ρ, with $N = \{8,16,32,64,128,181,256,512\}$ and with radii $\rho = \{0°,5°,12°,24°,48°,90°\}$. That makes (2000)(8)(6) = 96000 runs. For more details see the article[1].

Definitions:

probMIN0, probMAX0	probability distributions for alignment (MIN) and avoidance (MAX), functions of η, η_0, σ
probMIN, probMAX	same as above except these are functions of η and *N,* using η_0(N,c1,a1) and σ(N,c2,a2) to get η_0 and σ
signiMIN0, signiMAX0	significance as a function of (η, η_0, σ)
signiMIN, signiMAX	significance as a function of (η,N) using η_0(N,c1,a1) and σ(N,c2,a2) to get η_0 and σ
norm	a constant used to normalize the distribution (the integral of probability must be 1)
η	alignment angle
η0	"mean", a parameter with a value near the peak of the probability distribution
σ	"half-width", a parameter with a value near the distribution's half-width
c1MIN, a1MIN,...	parameters needed to find η0 and σ from the number of sources N.
c1MINplusMinus, ...	standard error (plus/minus) in parameters found in fitting random data
η0MIN, η0MAX	functions for finding the mean η0
σMIN, σMAX	functions for half-width σ

In[27]:=
```
(* y = ((η - η0)/σ) *)
(* dy = dη/σ *)
(* The normalization factor "norm" is needed for the probability density *)
norm = (NIntegrate[(1 + e^(4 (y-1)))^-1 e^(-y^2/2), {y, -∞, ∞}])^-1;
√(2π) norm (*Constant needed for Eq. (10) and (11) in the article¹.*)
```

Out[28]= 1.22029

In[29]:=
```
probMIN0[η_, η0_, σ_] :=
  norm/σ (1 + e^(4 (η-η0-σ)/σ))^-1 e^(-1/2 ((η - η0)/σ)^2) (*A Gaussian modified by an S-function (1+e^(4 (η-η0-σ)/σ))^-1.*)
```

In[30]:=
```
signiMIN0[η_, η0_, σ_] := NIntegrate[probMIN0[η1, η0, σ], {η1, -∞, η}]
```

Next, check that the normalization constant does not change from the alignment (MIN) case to the avoidance (MAX) case:

In[31]:=
```
normMAX = NIntegrate[(1 + e^(-4 (y+1)))^-1 e^(-y^2/2), {y, -∞, ∞}]^-1;
Print["The normalization constant for probMIN and probMAX are equal: ",
 normMAX, " and ", norm]
```

The normalization constant for probMIN and probMAX are equal: 0.486826 and 0.486826

In[33]:=
```
probMAX0[η_, η0_, σ_] := 1/(norm σ) (1 + e^(-4 (η-η0+σ)/σ))^-1 e^(-1/2 ((η - η0)/σ)^2)
```

In[34]:=
```
signiMAX0[η_, η0_, σ_] := NIntegrate[probMAX0[η1, η0, σ], {η1, η, ∞}]
```

The significance signiMIN0[η, η0, σ] is the integral of probMIN0, i.e. signiMIN0 = $\int_{-\infty}^{\eta} P_{MIN}(\eta)\, d\eta$.

The significance signiMAX0[η, η0, σ] is the integral of probMAX0, i.e. signiMAX0 = $\int_{\eta}^{\infty} P_{MAX}(\eta)\, d\eta$.

The formulas for mean $\eta_0 = \frac{\pi}{4} \pm \frac{c1}{N^{a1}}$ and half-width $\sigma = \frac{c2}{4\,N^{a2}}$ estimate η_0 and σ by functions of the number N of sources. These formulas depend on the size of the region (radius ρ) by the choice of parameters c_i and a_i, i = 1,2. The following values for the parameters c_i and a_i are based on random runs. For each combination of N = {8,16,32,64,128,181,256,512} and ρ = {0°,5°,12°,24°,48°,90°}, there were 2000 random runs completed.

A notation conflict between this notebook and the article[1] should be noted. We doubled the exponent “a” so $N^{a/2}$ appears in the article, whereas in the random runs and here we see N^a. Thus $a \approx 1/2$ here and in the random run fits, but the paper has $a_{\text{Article}} \approx 1$. That explains the “/2” in the following arrays.

```
In[35]:= ρciaiMIN = "ρ"  "c1"     "a1"      "c2"    "a2"
                    90  0.9423  1.0046/2  1.061  0.954/2
                    48  0.9505  1.0156/2  1.166  0.9956/2
                    24  0.9235  1.0069/2  1.127  0.964/2   ;
                    12  0.8912  1.0054/2  1.238  1.021/2
                    5   0.8363  1.0088/2  1.076  0.940/2
                    0   0.5031  1.0153/2  1.522  1.053/2
```

```
In[36]:= ρciaiMAX = "ρ"  "c1"     "a1"      "c2"    "a2"
                    90  0.9441  1.0055/2  1.000  0.931/2
                    48  0.9572  1.0165/2  1.090  0.958/2
                    24  0.927   1.0068/2  1.101  0.964/2;
                    12  0.9049  1.0090/2  1.228  1.018/2
                    5   0.8424  1.0062/2  1.168  0.992/2
                    0   0.4982  1.0093/2  1.543  1.060/2
```

```
In[37]:= ρΔciaiMIN = "ρ"  "c1"     "a1"      "c2"    "a2"
                     90  0.0050  0.0036/2  0.026  0.016/2
                     48  0.0079  0.0057/2  0.016  0.0095/2
                     24  0.0024  0.0018/2  0.022  0.013/2  ;
                     12  0.0034  0.0026/2  0.039  0.021/2
                     5   0.0035  0.0028/2  0.030  0.019/2
                     0   0.0059  0.0080/2  0.052  0.024/2
```

```
In[38]:= ρΔciaiMAX = "ρ"  "c1"     "a1"      "c2"    "a2"
                     90  0.0061  0.0044/2  0.038  0.025/2
                     48  0.0063  0.0045/2  0.026  0.016/2
                     24  0.011   0.0079/2  0.019  0.011/2;
                     12  0.0069  0.0052/2  0.039  0.022/2
                     5   0.0038  0.0031/2  0.022  0.013/2
                     0   0.0058  0.0080/2  0.057  0.025/2
```

If you have trouble translating the arrays from the pdf version into a viable Mathematica notebook, the following cells are equivalent. To activate a cell, remove the remark brackets (* and *).

```
In[39]:= (*ρciaiMIN={{"ρ","c1","a1","c2","a2"},
  {90,0.9423`,0.5023`,1.061`,0.477`},{48,0.9505`,0.5078`,1.166`,0.4978`},
  {24,0.9235`,0.50345`,1.127`,0.482`},{12,0.8912`,0.5027`,1.238`,0.5105`},
  {5,0.8363`,0.5044`,1.076`,0.47`},{0,0.5031`,0.50765`,1.522`,0.5265`}}*)
```

```
In[40]:= (*ρciaiMAX={{"ρ","c1","a1","c2","a2"},
  {90,0.9441`,0.50275`,1.`,0.4655`},{48,0.9572`,0.50825`,1.09`,0.479`},
  {24,0.927`,0.5034`,1.101`,0.482`},{12,0.9049`,0.5045`,1.228`,0.509`},
  {5,0.8424`,0.5031`,1.168`,0.496`},{0,0.4982`,0.50465`,1.543`,0.53`}}*)

In[41]:= (*ρΔciaiMIN={{"ρ","c1","a1","c2","a2"},
  {90,0.005`,0.0018`,0.026`,0.008`},{48,0.0079`,0.00285`,0.016`,0.00475`},
  {24,0.0024`,0.0009`,0.022`,0.0065`},{12,0.0034`,0.0013`,0.039`,0.0105`},
  {5,0.0035`,0.0014`,0.03`,0.0095`},{0,0.0059`,0.004`,0.052`,0.012`}}*)

In[42]:= (*ρΔciaiMAX={{"ρ","c1","a1","c2","a2"},
  {90,0.0061`,0.0022`,0.038`,0.0125`},{48,0.0063`,0.00225`,0.026`,0.008`},
  {24,0.011`,0.00395`,0.019`,0.0055`},{12,0.0069`,0.0026`,0.039`,0.011`},
  {5,0.0038`,0.00155`,0.022`,0.0065`},{0,0.0058`,0.004`,0.057`,0.0125`}}*)

In[43]:= (*Change the region radius, if necessary, in Section 3 Inputs and Settings. *)
iρ = regionChoice + 1; (* Parameters ci, ai, i = 1,2. *)
Print["These constants are for sources confined to regions with radii ρ = ",
 ρciaiMIN[[iρ, 1]], "°."]
{c1MIN, a1MIN, c2MIN, a2MIN} = Table[ρciaiMIN[[iρ, j]], {j, 2, 5}]
{c1MAX, a1MAX, c2MAX, a2MAX} = Table[ρciaiMAX[[iρ, j]], {j, 2, 5}]
Clear[iρ]

These constants are for sources confined to regions with radii ρ = 24°.

Out[44]= {0.9235, 0.50345, 1.127, 0.482}

Out[45]= {0.927, 0.5034, 1.101, 0.482}

In[47]:= (*Change the region radius, if necessary, in Section 3 Inputs and Settings. *)
iρ = regionChoice + 1; (* ± uncertainty for the parameters ci and ai, i = 1,2. *)
Print["These uncertainties are for sources confined to regions with radii ρ = ",
 ρciaiMAX[[iρ, 1]], "°."]
{c1MINplusMinus, a1MINplusMinus, c2MINplusMinus, a2MINplusMinus} =
 Table[ρΔciaiMIN[[iρ, j]], {j, 2, 5}]
{c1MAXplusMinus, a1MAXplusMinus, c2MAXplusMinus, a2MAXplusMinus} =
 Table[ρΔciaiMAX[[iρ, j]], {j, 2, 5}]
Clear[
 iρ]

These uncertainties are for sources confined to regions with radii ρ = 24°.

Out[48]= {0.0024, 0.0009, 0.022, 0.0065}

Out[49]= {0.011, 0.00395, 0.019, 0.0055}
```

In[51]:= η0MIN[nSrc_, c1_, a1_] := $\frac{\pi}{4} - \frac{c1}{nSrc^{a1}}$

σMIN[nSrc_, c2_, a2_] := $\frac{c2}{4\,nSrc^{a2}}$

In[53]:= η0MAX[nSrc_, c1_, a1_] := $\frac{\pi}{4} + \frac{c1}{nSrc^{a1}}$

σMAX[nSrc_, c2_, a2_] := $\frac{c2}{4\,nSrc^{a2}}$

The following probability distributions and significances make use of the above formulas for mean η_0 and half-width σ. They are functions of the alignment angle η and the number of sources N.

In[55]:=
```
probMIN[η_, nSrc_] := probMIN0[ η, η0MIN[nSrc, c1MIN, a1MIN], σMIN[nSrc, c2MIN, a2MIN] ]
```

In[56]:=
```
signiMIN[η_, nSrc_] := signiMIN0[η, η0MIN[nSrc, c1MIN, a1MIN], σMIN[nSrc, c2MIN, a2MIN]]
```

In[57]:=
```
probMAX[η_, nSrc_] := probMAX0[ η, η0MAX[nSrc, c1MAX, a1MAX], σMAX[nSrc, c2MAX, a2MAX] ]
signiMAX[η_, nSrc_] := signiMAX0[η, η0MAX[nSrc, c1MAX, a1MAX], σMAX[nSrc, c2MAX, a2MAX]]
```

5. Grid

We avoid bunching at the poles by taking into account the diminishing radii of constant latitude circles as the latitude approaches the poles. Successive grid points along any latitude or along any longitude make an arc that subtends the same central angle dθ.
We grid one hemisphere at a time, then they are combined.

Definitions:

gridSpacing	separation in degrees between grid points on a constant latitude circle and separation of constant latitude circles. Set by the user in Sec. 2.
dθ	grid spacing in radians
αpointH,δpointH	RA and dec of the grid points H_j
grid	see listing below for “grid” table entries
nGrid	number of grid points H_j , j = 1, 2, ..., nGrid
rGrid	radial unit vectors from origin to grid points, in 3D Cartesian coordinates
αGrid	RAs for grid points
δGrid	decs for grid points

Tables:

grid, gridN **and** ***gridS***

1. sequential point # 2. RA index 3. dec index 4. RA (rad) 5. dec (rad) 6. Cartesian coordinates of the grid point

In[59]:=
```
(*When gridSpacing = 2°, we get a 2°x2° grid.*)
Print["The grid spacing has been chosen in Sec. 3 to be gridSpacing = ", gridSpacing, "°."]
dθ = (2. π/360.) gridSpacing; (*Convert gridSpacing to radians*)
```

```
The grid spacing has been chosen in Sec. 3 to be gridSpacing = 2.°.
```

In[61]:=
```
(*The Northern Grid "gridN". *)
gridN = {}; idN = 1;

For[δj = 0., δj < π/(2. dθ), δj++, δpointH = δj dθ;
 For[ ai = 0., ai < Ceiling[(2. π)/dθ (Cos[δpointH] + 0.01)], ai++, αpointH = ai dθ/(Cos[δpointH] + 0.01);
  AppendTo[gridN, {idN, ai, δj, αpointH, δpointH, er[αpointH, δpointH]}];
  idN = idN + 1
 ]]
```

In[63]:=
```
(*The Southern Grid "gridS". *)
gridS = {}; idS = 1;

For[δj = 1., δj < π/(2. dθ), δj++, δpointH = -δj dθ;
 (*Print["{δj,δpointH} = ",{δj,δpointH}];*)
 For[ ai = 0., ai < Ceiling[(2. π)/dθ (Cos[δpointH] + 0.01)], ai++, αpointH = ai dθ/(Cos[δpointH] + 0.01);
  (*Print["{ai,αpointH} = ",{ai,αpointH}];*)
  AppendTo[gridS, {idS, ai, δj, αpointH, δpointH, er[αpointH, δpointH]}];
  idS = idS + 1
 ]]
```

In[65]:=
```
grid = {}; j = 1;
For[jN = 1, jN ≤ Length[gridN], jN++, AppendTo[grid,
   {j, gridN[[jN, 2]], gridN[[jN, 3]], gridN[[jN, 4]], gridN[[jN, 5]], gridN[[jN, 6]]}];
 j = j + 1]
For[jS = 1, jS ≤ Length[gridS], jS++, AppendTo[grid,
   {j, gridS[[jS, 2]], gridS[[jS, 3]], gridS[[jS, 4]], gridS[[jS, 5]], gridS[[jS, 6]]}];
 j = j + 1]
nGrid = Length[grid];
```

In[69]:=
```
αGrid = Table[αFROMr[grid[[j, 6]] ], {j, Length[grid]}];
δGrid = Table[δFROMr[grid[[j, 6]] ], {j, Length[grid]}];
rGrid = Table[grid[[j, 6]] , {j, Length[grid]}];
```

In[72]:=
```
Print["There are ", nGrid, " points on the grid. "]
```

```
There are 10518 points on the grid.
```

6. Analysis

Definitions:

vψSrc	unit vectors along the polarization directions in the tangent planes of the sources
jηBarHj	$\{j,\overline{\eta}(\text{H})\}$, where j is the index for grid point H_j and $\overline{\eta}(\text{H})$ is the average alignment angle at H_j. See Eq. (1) in the Introduction.
sortjηBarHj	$\{j,\overline{\eta}(\text{H})\}$, rearranged by value of $\overline{\eta}(\text{H})$, with smallest angles $\overline{\eta}(\text{H})$ first.

jηBarMin	$\{j,\overline{\eta}(H)\}$, the j and $\overline{\eta}$ for the smallest value of $\overline{\eta}(H)$, best alignment
ηBarMin	the smallest value of $\overline{\eta}(H)$, measures alignment of the polarization directions
jηBarMax	$\{j,\overline{\eta}(H)\}$, the j and $\overline{\eta}$ for the largest value of $\overline{\eta}(H)$, most avoided
ηBarMax	the largest value of $\overline{\eta}(H)$, measures avoidance
sigηBarMin	significance of the smallest alignment angle
sigRangeηBarMin	using the plus/minus values on the parameters c_i and a_i, the table collects corresponding values of the significance
sigSmallηBarMin	the smallest of the values in sigRangeηBarMin
sigBigηBarMin	the largest of the values in sigRangeηBarMin
sigηBarMax	significance of the largest alignment angle (i.e. avoidance)
sigRangeηBarMax	using the plus/minus values on the parameters c_i and a_i, the table collects corresponding values of the significance
sigSmallηBarMax	the smallest of the values in sigRangeηBarMax
sigBigηBarMax	the largest of the values in sigRangeηBarMax
αHminDegrees	RA of the point $H_{min}$where $\overline{\eta}(H)$ is the smallest
δHminDegrees	dec of the point $H_{min}$where $\overline{\eta}(H)$ is the smallest
αHmaxDegrees	RA of the point $H_{max}$where $\overline{\eta}(H)$ is the largest
δHmaxDegrees	dec of the point $H_{max}$where $\overline{\eta}(H)$ is the largest

In[73]:=
```
(*Analysis using Eq (5) in the article¹ to get ηiH, cos(η) = |v̂H.v̂ψ|,
then {j,η̄(Hj)}, which are sorted to get the extreme values*)
vψSrc = Table[Cos[ ψn[[i]] ] eN[ αSrc[[i]], δSrc[[i]] ] +
   Sin[ ψn[[i]] ] eE[ αSrc[[i]], δSrc[[i]] ], {i, nSrc}];
jηBarHj = Table[{j, (1/nSrc) Sum[ArcCos[ Abs[ rGrid[[j]].vψSrc[[i]] /
        ((rGrid[[j]] - (rGrid[[j]].rSrc[[i]]) rSrc[[i]]).(rGrid[[j]] - (rGrid[[j]].
                rSrc[[i]]) rSrc[[i]]))^(1/2) ] - 0.000001 ] , {i, nSrc}]}, {j, nGrid}];
 sortjηBarHj = Sort[jηBarHj, #1[[2]] < #2[[2]] &]; jηBarMin = sortjηBarHj[[1]];
ηBarMin = jηBarMin[[2]];
jηBarMax = sortjηBarHj[[-1]];
ηBarMax = jηBarMax[[2]] ;
```

In[78]:=
```
(*Alternate analysis using Eq (7) in the article¹ to get ηiH, cos(η) = |n̂Sxψ.n̂SxH|.*)
(*nSxψn  =  Table[ Sin[ψn[[n]]]eN[αSrc[[n]],δSrc[[n]]]-
   Cos[ψn[[n]]]eE[αSrc[[n]],δSrc[[n]]],    {n,nSrc}];
nSxHnj[j_]:=nSxHnj[j]=Table[ Cross[  rSrc[[n]],rGrid[[j]]  ]/
    (√((Cross[  rSrc[[n]],rGrid[[j]]  ]).(Cross[  rSrc[[n]],rGrid[[j]]  ]))) ,    {n,
    nSrc}];
ηnHj[j_]:=ηnHj[j]=Table[ ArcCos[  Abs[  nSxψn[[n]].nSxHnj[j][[n]]  ] -
     0.000001  ],      {n,nSrc}];
ηBarHj[j_]:=ηBarHj[j]=Sum[ηnHj[j][[n]],{n,nSrc}]/nSrc
   jηBarHj=Table[{j,ηBarHj[j]},{j,Length[grid]}];
sortjηBarHj=Sort[jηBarHj,#1[[2]]<#2[[2]]&];
jηBarMin=sortjηBarHj[[1]];
ηBarMin=jηBarMin[[2]]
  jηBarMax=sortjηBarHj[[-1]];
ηBarMax=jηBarMax[[2]]*)
```

In[79]:=
```
(*Significance of the alignment of the polarization directions with hub point Hmin.*)
sigηBarMin = signiMIN[ηBarMin, nSrc];
sigRangeηBarMin = Sort[Partition[Flatten[Table[
      {signiMIN0[ηBarMin, η0MIN[nSrc, c1MIN + γ1 c1MINplusMinus, a1MIN + α1 a1MINplusMinus],
        σMIN[nSrc, c2MIN + γ2 c2MINplusMinus, a2MIN + α2 a2MINplusMinus]], γ1, α1, γ2, α2},
      {γ1, -1, 1}, {α1, -1, 1}, {γ2, -1, 1}, {α2, -1, 1}] ], 5 ]   ];
{sigRangeηBarMin[[1]], sigRangeηBarMin[[-1]]};
sigSmallηBarMin = sigRangeηBarMin[[1, 1]];
sigBigηBarMin = sigRangeηBarMin[[-1, 1]];
Print["The best value for the significance of alignment is sig. =  ", sigηBarMin,
 ". Using the uncertainties +/- of the ci,ai, the lowest and highest values are ",
 sigSmallηBarMin, " and ", sigBigηBarMin  , " giving the range from sig. = ",
 sigSmallηBarMin, " to ", sigBigηBarMin, " . "]
```

The best value for the significance of alignment is sig. = 0.0111662
. Using the uncertainties +/- of the c_i,a_i, the lowest and highest values are
0.00832443 and 0.0146188 giving the range from sig. = 0.00832443 to 0.0146188 .

In[85]:=
```
(*Significance of the polarization directions' avoidance of the hub point Hmax.*)
sigηBarMax = signiMAX[ηBarMax, nSrc];
sigRangeηBarMax = Sort[Partition[Flatten[Table[
      {signiMAX0[ηBarMax, η0MAX[nSrc, c1MAX + γ1 c1MAXplusMinus, a1MAX + α1 a1MAXplusMinus],
        σMAX[nSrc, c2MAX + γ2 c2MAXplusMinus, a2MAX + α2 a2MAXplusMinus]], γ1, α1, γ2, α2},
      {γ1, -1, 1}, {α1, -1, 1}, {γ2, -1, 1}, {α2, -1, 1}] ], 5 ]   ];
{sigRangeηBarMax[[1]], sigRangeηBarMax[[-1]]};
sigSmallηBarMax = sigRangeηBarMax[[1, 1]];
sigBigηBarMax = sigRangeηBarMax[[-1, 1]];
Print["The best value for the significance of avoidance is sig. =  ", sigηBarMax,
 ". Using the uncertainties +/- of the ci,ai, the lowest and highest values are ",
 sigSmallηBarMax, " and ", sigBigηBarMax  , " giving the range from sig. = ",
 sigSmallηBarMax, " to ", sigBigηBarMax, " . "]
```

The best value for the significance of avoidance is sig. = 0.0268444
. Using the uncertainties +/- of the c_i,a_i, the lowest and highest values are
0.016778 and 0.0411734 giving the range from sig. = 0.016778 to 0.0411734 .

In[91]:=
```
{jηBarMin, jηBarMax};(* {1. grid#, 2. alignment angle η} at Min and Max η .*)
αHminDegrees0 = grid[[ jηBarMin[[1]] ]][[4]] (360 / (2 π));
δHminDegrees0 = grid[[ jηBarMin[[1]] ]][[5]] (360 / (2 π));
If[(180 < αHminDegrees0 < 361), αHminDegrees = αHminDegrees0 - 180;
  δHminDegrees = -δHminDegrees0 , αHminDegrees = αHminDegrees0;
  δHminDegrees = δHminDegrees0];
αHmaxDegrees0 = grid[[ jηBarMax[[1]] ]][[4]] (360 / (2 π));
δHmaxDegrees0 = grid[[ jηBarMax[[1]] ]][[5]] (360 / (2 π));
If[(180 < αHmaxDegrees0 < 361), αHmaxDegrees = αHmaxDegrees0 - 180;
  δHmaxDegrees = -δHmaxDegrees0 , αHmaxDegrees = αHmaxDegrees0;
  δHmaxDegrees = δHmaxDegrees0];
Print["The alignment hub Hmin is located at (RA,dec) = ", {αHminDegrees, δHminDegrees },
 " and at ", {αHminDegrees - 180, -δHminDegrees }, " , in degrees"]
Print["The avoidance hub Hmax is located at (RA,dec) = ", {αHmaxDegrees, δHmaxDegrees },
 " and at ", {αHmaxDegrees - 180, -δHmaxDegrees }, " , in degrees"]
```

The alignment hub H_{min} is located at (RA,dec) = {106.408, -20.} and at {-73.5915, 20.} , in degrees

The avoidance hub H_{max} is located at (RA,dec) = {9.93072, -22.} and at {-170.069, 22.} , in degrees

7. Plot of the alignment function $\overline{\eta}$(H)

Definitions

αjδjηBarHjTable	{RA_j, dec_j, $\overline{\eta}$(H)} at each grid point $H = H_j$, in degrees
ηBarHjSmooth	interpolation of αjδjηBarHjTable yields $\overline{\eta}$(H) as a smooth function of the (RA,dec) of H
xyηBarAitoffTable	{x, y, $\overline{\eta}$(x,y)} , where x,y are Aitoff coordinates and $\overline{\eta}$(x,y) is the alignment angle
dηContourPlot	separation of successive contour lines, in degrees
listCP	list contour plot of $\overline{\eta}$(H), from xyηBarAitoffTable
xyAitoffSources	{x,y} Aitoff coordinates for the sources' locations on the sphere
mapOfηAitoff	contour plot listCP of the alignment angle $\overline{\eta}$(H) , with source locations and labels

αH(α,δ) , xH(α,δ) , yH(α,δ) are functions needed when making a 2-D map of the Celestial sphere. The origin xH, yH is centered on $\alpha = \delta = 0$. Notice the naming conflict: αH(α,δ) is an Aitoff parameter which, in general, differs from the Right Ascension α .

In[100]:=
```
(*The following table αjδjηBarHjTable is interpolated below
 to yield a smooth function of the alignment angle over the sphere.*)
(* Table Entries: 1.  RA at jth grid point (degrees) 2. dec at jth grid
   point (degrees)  3. alignment angle ηBarRgnkj at jth grid point (degrees)*)
αjδjηBarHjTable = (αjδjηBarHjTable0 = {};
   For[j = 1, j ≤ Length[jηBarHj], j++,
    AppendTo[ αjδjηBarHjTable0,  {grid[[j, 4]] * (360. / (2. π)), grid[[j, 5]] * (360. / (2. π)),
      jηBarHj[[j, 2]] * (360. / (2. π))}]   ; If[ 360 ≥ grid[[j, 4]] * (360. / (2. π)) > 354.,
     AppendTo[ αjδjηBarHjTable0,  {grid[[j, 4]] * (360. / (2. π)) - 360.,
       grid[[j, 5]] * (360. / (2. π)), jηBarHj[[j, 2]] * (360. / (2. π))}]   ] ;
    If[ 6. > grid[[j, 4]] * (360. / (2. π)) ≥ 0., AppendTo[ αjδjηBarHjTable0,
      {grid[[j, 4]] * (360. / (2. π)) + 360, grid[[j, 5]] * (360. / (2. π)),
       jηBarHj[[j, 2]] * (360. / (2. π))}]   ]        ];
   αjδjηBarHjTable0);
```

In[101]:=
```
ηBarHjSmooth = Interpolation[αjδjηBarHjTable]
 (*The smooth alignment angle function for the region.*)
```

Interpolation: Interpolation on unstructured grids is currently only supported for InterpolationOrder->1 or InterpolationOrder->All. Order will be reduced to 1.

Out[101]=
```
InterpolatingFunction[ Domain: {{-5.92, 366.}, {-88., 88.}}
                       Output: scalar ]
```

The following Aitoff Plot formulas[3] were be found in, for example, Wikipedia contributors. "Aitoff projection." Wikipedia, The Free Encyclopedia. Wikipedia, The Free Encyclopedia, 25 May. 2017. Web. 3 Jan. 2018.

In[102]:=
```
αH[α_, δ_] := αH[α, δ] = ArcCos[Cos[2. π/360. δ] Cos[2. π/360. α/2.]] (*angles α and δ are in degrees*)
xH[α_, δ_] := xH[α, δ] = (2. Cos[2. π/360. δ] Sin[2. π/360. α/2.])/Sinc[αH[α, δ]]
yH[α_, δ_] := yH[α, δ] = Sin[2. π/360. δ]/Sinc[αH[α, δ]]
```

In[105]:=
```
xyηBarAitoffTable = Partition[Flatten[Table[{xH[α - 180, -δ], yH[α - 180, -δ], ηBarHjSmooth[α, δ]},
    {α, 0, 360., 2.}, {δ, -88., 88., 2.}]], 3];
(* The smooth alignment angle function ηBarHjSmooth mapped onto a 2D
 Aitoff projection of the sphere. *)

xyAitoffSources =
  Table[{xH[ αSrc[[n]] 360/(2 π), δSrc[[n]] 360/(2 π) ], yH[ αSrc[[n]] 360/(2 π), δSrc[[n]] 360/(2 π) ]}, {n, nSrc}];
 (*The Aitoff coordinates for the sources' locations.*)
xyAitoffOppositeSources =
  Table[{xH[ If[0 < αSrc[[n]] 360/(2 π) < +180, αSrc[[n]] 360/(2 π) - 180, If[0 > αSrc[[n]] 360/(2 π) > -180,
       αSrc[[n]] 360/(2 π) + 180]], -δSrc[[n]] 360/(2 π) ], yH[ If[0 < αSrc[[n]] 360/(2 π) < +180, αSrc[[n]] 360/(2 π) -
       180, If[0 > αSrc[[n]] 360/(2 π) > -180, αSrc[[n]] 360/(2 π) + 180]], -δSrc[[n]] 360/(2 π) ]}, {n, nSrc}];
```

In[108]:=
```
(* Contour plot of the alignment function ηBarHjSmooth. *)
listCP = ListContourPlot[Union[xyηBarAitoffTable (*,
     {{xH[αHminDegrees,δHminDegrees],yH[αHminDegrees,δHminDegrees],ηBarMin*(360./(2.π))-1.0}},
     {{xH[αHmaxDegrees,δHmaxDegrees],yH[αHmaxDegrees,δHmaxDegrees],
       ηBarMax*(360./(2.π))+1.0}}*)], AspectRatio → 1/2,
    Contours → Table[η, {η, Floor[jηBarMin[[2]] * (360. / (2. π))] + 1,
       Ceiling[jηBarMax[[2]] * (360. / (2. π))] - 1, dηContourPlot}],
    ColorFunction → "TemperatureMap", PlotRange → {{-7, 7}, {-3, 3}}, Axes -> False, Frame → False] ;
```

```
In[109]:= (*Construct the map of η̄(H).*)
Print["The map is centered on (RA,dec) = (0°,0°)."]
Print["The map is symmetric across diameters, i.e.
   diametrically opposite points -H and H have the same alignment angle."]
Print["The contour lines are separated by ", dηContourPlot,
  "°. This choice can be reset in Sec. 3."]
Print["Source dots are Purple, the dots opposite the sources are Magenta."]
Print["The best alignment angle (min) is η̄min = ", jηBarMin[[2]] (360. / (2. π)), "°."]
Print["The best avoidance angle (max) is η̄max = ", jηBarMax[[2]] (360. / (2. π)), "°."]
Print["The alignment hubs Hmin and -Hmin are located at (RA,dec) = ",
  {αHminDegrees, δHminDegrees }, " and at ", {αHminDegrees - 180, -δHminDegrees }, " , in degrees."]
Print["The avoidance hubs Hmax and -Hmax are located at (RA,dec) = ",
  {αHmaxDegrees, δHmaxDegrees }, " and at ", {αHmaxDegrees - 180, -δHmaxDegrees }, " , in degrees."]
mapOfηAitoff =
 Show[{listCP,
   Table[ParametricPlot[{xH[α, δ], yH[α, δ]}, {δ, -90, 90}, PlotStyle → {Black, Thickness[0.002]},
     (*Mesh→{11,5,0}(*{23,11,0}*),MeshStyle→Thick,*)PlotPoints → 60], {α, -180, 180, 30}],
   Table[ParametricPlot[{xH[α, δ], yH[α, δ]}, {α, -180, 180},
     PlotStyle → {Black, Thickness[0.002]}, (*Mesh→{11,5,0}
     (*{23,11,0}*),MeshStyle→Thick,*)PlotPoints → 60], {δ, -60, 60, 30}],
   Graphics[{PointSize[0.007], Text[StyleForm["N", FontSize -> 10, FontWeight -> "Plain"],
      {0, 1.85}], (*Sources S:*)Purple, Point[ xyAitoffSources ],
     (*Opposite from sources, -S:*)Magenta, Point[xyAitoffOppositeSources],
      Black, Text[StyleForm["Max", FontSize → 8, FontWeight -> "Bold"],
      {xH[ - 180 , 0], yH[0, -60]}], {Arrow[BezierCurve[{{xH[ - 165 , 0], yH[0, -50]}, {-2.5, -1.2},
         {xH[αHmaxDegrees - 180, -δHmaxDegrees], yH[αHmaxDegrees - 180, -δHmaxDegrees]}}]]},
    Text[StyleForm["Min", FontSize → 8, FontWeight -> "Bold"], {xH[ 180 , 0], yH[0, -60]}],
    {Arrow[BezierCurve[{{xH[ 165 , 0], yH[0, -50]}, {1.6, -1.20},
         {xH[αHminDegrees, δHminDegrees], yH[αHminDegrees, δHminDegrees]}}]]},
    Text[StyleForm["Min", FontSize → 8, FontWeight -> "Bold"], {xH[ -180 , 0], yH[0, 60]}],
    {Arrow[BezierCurve[{{xH[ -165 , 0], yH[0, 50]}, {-2.4, 1.8},
         {xH[αHminDegrees - 180, -δHminDegrees], yH[αHminDegrees - 180, -δHminDegrees]}}]]},
    Text[StyleForm["Max", FontSize → 8, FontWeight -> "Bold"], {xH[ 180 , 0], yH[0, 60]}] ,
    {Arrow[BezierCurve[{{xH[ 165 , 0], yH[0, 50]}, {2.2, 0.0}, {xH[αHmaxDegrees, δHmaxDegrees],
          yH[αHmaxDegrees, δHmaxDegrees]}}]]}      }]}, ImageSize → 432]
```

The map is centered on (RA,dec) = (0°,0°).

The map is symmetric across diameters, i.e.
diametrically opposite points -H and H have the same alignment angle.

The contour lines are separated by 4°. This choice can be reset in Sec. 3.

Source dots are Purple, the dots opposite the sources are Magenta.

The best alignment angle (min) is $\overline{\eta}_{min}$ = 21.8882°.

The best avoidance angle (max) is $\overline{\eta}_{max}$ = 68.769°.

The alignment hubs H_{min} and $-H_{min}$ are located at (RA,dec) = {106.408, -20.} and at {-73.5915, 20.} , in degrees.

The avoidance hubs H_{max} and $-H_{max}$ are located at (RA,dec) = {9.93072, -22.} and at {-170.069, 22.} , in degrees.

Out[117]=



In[118]:=
```
Print["The number of sources: N = ", nSrc]
Print["The min alignment angle is ηmin = ", jηBarMin[[2]] * (360. / (2. π)),
 "° , which has a significance of sig. = ", sigηBarMin, ",  plus/minus = + ",
 sigBigηBarMin - sigηBarMin, " and - ", sigηBarMin - sigSmallηBarMin,
 " , giving a range from sig. = ", sigSmallηBarMin, " to ", sigBigηBarMin, " ."]
Print["The max avoidance angle is ηmax = ", jηBarMax[[2]] * (360. / (2. π)),
 "° , which has a significance of sig. = ", sigηBarMax, ",  plus/minus = + ",
 sigBigηBarMax - sigηBarMax, " and - ", sigηBarMax - sigSmallηBarMax,
 " , giving a range from sig. = ", sigSmallηBarMax, " to ", sigBigηBarMax, " ."]
```

The number of sources: N = 16

The min alignment angle is ηmin = 21.8882
° , which has a significance of sig. = 0.0111662, plus/minus = + 0.0034526
and - 0.00284176 , giving a range from sig. = 0.00832443 to 0.0146188 .

The max avoidance angle is ηmax = 68.769° , which has a significance of sig. = 0.0268444
, plus/minus = + 0.014329 and - 0.0100664 , giving a range from sig. = 0.016778 to 0.0411734 .

In[121]:=
```
(*Export the map "mapOfηAitoff" as a pdf. The export location can be reset in Sec. 3.*)
(*To activate, remove the remark brackets "(*" and "*)". *)
(*SetDirectory[mapDirectory];
Export["mapForStarterKit.pdf",
 Show[mapOfηAitoff,ImageSize→432],"PDF",ImageSize→{480,Automatic}]*)
```

References

0.* R. Shurtleff, the viable Mathematica notebook is available, for a limited time, at the following URL:
https://www.dropbox.com/s/31f8jak2e73re3c/20201211StarterKitForHubTest4.nb?dl=0 (2021)
1. R. Shurtleff, "Indirect polarization alignment with points on the sky, the Hub Test" , https://vixra.org/abs/2011.0026 (2020).
2. Wolfram Research, Inc., Mathematica, Version 12.1, Champaign, IL (2020).
3. Wikipedia contributors. "Aitoff projection." Wikipedia, The Free Encyclopedia. Wikipedia, The Free Encyclopedia, 25 May. 2017. Web. 3 Jan. 2018.

*The online version has

0.* R. Shurtleff, the viable Mathematica notebook is available, for a limited time, at the following URL: (Check the references in the pdf version. The URL cannot appear here now. I blame the Principle of Causality. The plan is to include the URL once the URL is created. Time will tell.)